pragma solidity ^0.4.18; // solhint-disable-line



/// @title Interface for contracts conforming to ERC-721: Non-Fungible Tokens
/// @author Dieter Shirley <dete@axiomzen.co> (https://github.com/dete)
contract ERC721 {
  // Required methods
  function approve(address _to, uint256 _tokenId) public;
  function balanceOf(address _owner) public view returns (uint256 balance);
  function implementsERC721() public pure returns (bool);
  function ownerOf(uint256 _tokenId) public view returns (address addr);
  function takeOwnership(uint256 _tokenId) public;
  function totalSupply() public view returns (uint256 total);
  function transferFrom(address _from, address _to, uint256 _tokenId) public;
  function transfer(address _to, uint256 _tokenId) public;

  event Transfer(address indexed from, address indexed to, uint256 tokenId);
  event Approval(address indexed owner, address indexed approved, uint256 tokenId);

  // Optional
  // function name() public view returns (string name);
  // function symbol() public view returns (string symbol);
  // function tokenOfOwnerByIndex(address _owner, uint256 _index) external view returns (uint256 tokenId);
  // function tokenMetadata(uint256 _tokenId) public view returns (string infoUrl);
}


contract GameItemNew is ERC721 {

  address cryptoVideoGames = 0xdec14d8f4da25108fd0d32bf2decd9538564d069; 

  /*** EVENTS ***/

  /// @dev The Birth event is fired whenever a new Game Item comes into existence.
  event Birth(uint256 tokenId, string name, address owner);

  /// @dev The TokenSold event is fired whenever a token is sold.
  event TokenSold(uint256 tokenId, uint256 oldPrice, uint256 newPrice, address prevOwner, address winner, string name);

  /// @dev Transfer event as defined in current draft of ERC721. 
  ///  ownership is assigned, including births.
  event Transfer(address from, address to, uint256 tokenId);

  /*** CONSTANTS ***/

  /// @notice Name and symbol of the non fungible token, as defined in ERC721.
  string public constant NAME = "CryptoKotakuGameItemNew"; // solhint-disable-line
  string public constant SYMBOL = "GameItemNew"; // solhint-disable-line

  uint256 private startingPrice = 0.005 ether;

  /*** STORAGE ***/

  /// @dev A mapping from Game Item IDs to the address that owns them. All Game Items have
  ///  some valid owner address.
  mapping (uint256 => address) public gameItemIndexToOwner;

  // @dev A mapping from owner address to count of tokens that address owns.
  //  Used internally inside balanceOf() to resolve ownership count.
  mapping (address => uint256) private ownershipTokenCount;

  /// @dev A mapping from gameItemIDs to an address that has been approved to call
  ///  transferFrom(). Each game Item can only have one approved address for transfer
  ///  at any time. A zero value means no approval is outstanding.
  mapping (uint256 => address) public gameItemIndexToApproved;

  // @dev A mapping from game Item IDs to the price of the token.
  mapping (uint256 => uint256) private gameItemIndexToPrice;

  // The addresses of the accounts (or contracts) that can execute actions within each roles.
  address public ceoAddress;
  address public cooAddress;

  uint256 public promoCreatedCount;

  /*** DATATYPES ***/
  struct GameItem {
    string name;
    uint gameId;
  }

  GameItem[] private gameItems;

  /*** ACCESS MODIFIERS ***/
  /// @dev Access modifier for CEO-only functionality
  modifier onlyCEO() {
    require(msg.sender == ceoAddress);
    _;
  }

  /// @dev Access modifier for COO-only functionality
  modifier onlyCOO() {
    require(msg.sender == cooAddress);
    _;
  }

  /// Access modifier for contract owner only functionality
  modifier onlyCLevel() {
    require(
      msg.sender == ceoAddress ||
      msg.sender == cooAddress
    );
    _;
  }

  /*** CONSTRUCTOR ***/
  function GameItemNew() public {
    ceoAddress = msg.sender;
    cooAddress = msg.sender;
  }

  /*** PUBLIC FUNCTIONS ***/
  /// @notice Grant another address the right to transfer token via takeOwnership() and transferFrom().
  /// @param _to The address to be granted transfer approval. Pass address(0) to
  ///  clear all approvals.
  /// @param _tokenId The ID of the Token that can be transferred if this call succeeds.
  /// @dev Required for ERC-721 compliance.
  function approve(
    address _to,
    uint256 _tokenId
  ) public {
    // Caller must own token.
    require(_owns(msg.sender, _tokenId));

    gameItemIndexToApproved[_tokenId] = _to;

    Approval(msg.sender, _to, _tokenId);
  }

  /// For querying balance of a particular account
  /// @param _owner The address for balance query
  /// @dev Required for ERC-721 compliance.
  function balanceOf(address _owner) public view returns (uint256 balance) {
    return ownershipTokenCount[_owner];
  }

  /// @dev Creates a new promo game Item with the given name, with given _price and assignes it to an address.
  function createPromoGameItem(address _owner, string _name, uint256 _price, uint _gameId) public onlyCOO {

    address gameItemOwner = _owner;
    if (gameItemOwner == address(0)) {
      gameItemOwner = cooAddress;
    }

    if (_price <= 0) {
      _price = startingPrice;
    }

    promoCreatedCount++;
    _createGameItem(_name, gameItemOwner, _price, _gameId);
  }

  /// @dev Creates a new gameItem with the given name.
  function createContractGameItem(string _name, uint _gameId) public onlyCOO {
    _createGameItem(_name, address(this), startingPrice, _gameId);
  }

  /// @notice Returns all the relevant information about a specific GameItem.
  /// @param _tokenId The tokenId of the GameItem of interest.
  function getGameItem(uint256 _tokenId) public view returns (
    uint256 Id,
    string gameItemName,
    uint256 sellingPrice,
    address owner,
    uint gameId
  ) {
    GameItem storage gameItem = gameItems[_tokenId];
    Id = _tokenId;
    gameItemName = gameItem.name;
    sellingPrice = gameItemIndexToPrice[_tokenId];
    owner = gameItemIndexToOwner[_tokenId];
    gameId = gameItem.gameId;
  }

  function implementsERC721() public pure returns (bool) {
    return true;
  }

  /// @dev Required for ERC-721 compliance.
  function name() public pure returns (string) {
    return NAME;
  }

  /// For querying owner of token
  /// @param _tokenId The tokenID for owner inquiry
  /// @dev Required for ERC-721 compliance.
  function ownerOf(uint256 _tokenId)
    public
    view
    returns (address owner)
  {
    owner = gameItemIndexToOwner[_tokenId];
    require(owner != address(0));
  }

  function payout(address _to) public onlyCLevel {
    _payout(_to);
  }

  // Allows someone to send ether and obtain the token
  function purchase(uint256 _tokenId) public payable {
    address oldOwner = gameItemIndexToOwner[_tokenId];
    address newOwner = msg.sender;

    uint256 sellingPrice = gameItemIndexToPrice[_tokenId];

    // Making sure token owner is not sending to self
    require(oldOwner != newOwner);

    // Safety check to prevent against an unexpected 0x0 default.
    require(_addressNotNull(newOwner));

    // Making sure sent amount is greater than or equal to the sellingPrice
    require(msg.value >= sellingPrice);

    uint256 gameOwnerPayment = uint256(SafeMath.div(SafeMath.mul(sellingPrice, 10), 100));
    uint256 devFees = uint256(SafeMath.div(SafeMath.mul(sellingPrice, 6), 100));
    uint256 payment =  sellingPrice - devFees - gameOwnerPayment;
    uint256 purchaseExcess = SafeMath.sub(msg.value,sellingPrice);

   
    gameItemIndexToPrice[_tokenId] = SafeMath.div(SafeMath.mul(sellingPrice, 150), 100);
    
    _transfer(oldOwner, newOwner, _tokenId);
    TokenSold(_tokenId, sellingPrice, gameItemIndexToPrice[_tokenId], oldOwner, newOwner, gameItems[_tokenId].name);

    // Pay previous tokenOwner if owner is not contract
    if (oldOwner != address(this)) {
      oldOwner.transfer(payment);
    }
    
    msg.sender.transfer(purchaseExcess);
    _transferDivs(gameOwnerPayment, _tokenId, devFees);
    
  }

  /// Divident distributions
  function _transferDivs(uint256 _gameOwnerPayment, uint256 _tokenId,uint256 _devFees) private {
    CryptoVideoGames gamesContract = CryptoVideoGames(cryptoVideoGames);
    address gameOwner = gamesContract.getVideoGameOwner(gameItems[_tokenId].gameId);
    gameOwner.transfer(_gameOwnerPayment);
    ceoAddress.transfer(_devFees);
  }

  function priceOf(uint256 _tokenId) public view returns (uint256 price) {
    return gameItemIndexToPrice[_tokenId];
  }

  /// @dev Assigns a new address to act as the CEO. Only available to the current CEO.
  /// @param _newCEO The address of the new CEO
  function setCEO(address _newCEO) public onlyCEO {
    require(_newCEO != address(0));

    ceoAddress = _newCEO;
  }

  /// @dev Assigns a new address to act as the COO. Only available to the current COO.
  /// @param _newCOO The address of the new COO
  function setCOO(address _newCOO) public onlyCEO {
    require(_newCOO != address(0));

    cooAddress = _newCOO;
  }

  /// @dev Required for ERC-721 compliance.
  function symbol() public pure returns (string) {
    return SYMBOL;
  }

  /// @notice Allow pre-approved user to take ownership of a token
  /// @param _tokenId The ID of the Token that can be transferred if this call succeeds.
  /// @dev Required for ERC-721 compliance.
  function takeOwnership(uint256 _tokenId) public {
    address newOwner = msg.sender;
    address oldOwner = gameItemIndexToOwner[_tokenId];

    // Safety check to prevent against an unexpected 0x0 default.
    require(_addressNotNull(newOwner));

    // Making sure transfer is approved
    require(_approved(newOwner, _tokenId));

    _transfer(oldOwner, newOwner, _tokenId);
  }

  /// @param _owner The owner whose gameItem tokens we are interested in.
  /// @dev This method MUST NEVER be called by smart contract code. First, it's fairly
  ///  expensive (it walks the entire gameItems array looking for gameItems belonging to owner),
  ///  but it also returns a dynamic array, which is only supported for web3 calls, and
  ///  not contract-to-contract calls.
  function tokensOfOwner(address _owner) public view returns(uint256[] ownerTokens) {
    uint256 tokenCount = balanceOf(_owner);
    if (tokenCount == 0) {
        // Return an empty array
      return new uint256[](0);
    } else {
      uint256[] memory result = new uint256[](tokenCount);
      uint256 totalGameItems = totalSupply();
      uint256 resultIndex = 0;

      uint256 gameItemId;
      for (gameItemId = 0; gameItemId <= totalGameItems; gameItemId++) {
        if (gameItemIndexToOwner[gameItemId] == _owner) {
          result[resultIndex] = gameItemId;
          resultIndex++;
        }
      }
      return result;
    }
  }

  /// For querying totalSupply of token
  /// @dev Required for ERC-721 compliance.
  function totalSupply() public view returns (uint256 total) {
    return gameItems.length;
  }

  /// Owner initates the transfer of the token to another account
  /// @param _to The address for the token to be transferred to.
  /// @param _tokenId The ID of the Token that can be transferred if this call succeeds.
  /// @dev Required for ERC-721 compliance.
  function transfer(
    address _to,
    uint256 _tokenId
  ) public {
    require(_owns(msg.sender, _tokenId));
    require(_addressNotNull(_to));

    _transfer(msg.sender, _to, _tokenId);
  }

  /// Third-party initiates transfer of token from address _from to address _to
  /// @param _from The address for the token to be transferred from.
  /// @param _to The address for the token to be transferred to.
  /// @param _tokenId The ID of the Token that can be transferred if this call succeeds.
  /// @dev Required for ERC-721 compliance.
  function transferFrom(
    address _from,
    address _to,
    uint256 _tokenId
  ) public {
    require(_owns(_from, _tokenId));
    require(_approved(_to, _tokenId));
    require(_addressNotNull(_to));

    _transfer(_from, _to, _tokenId);
  }

  /*** PRIVATE FUNCTIONS ***/
  /// Safety check on _to address to prevent against an unexpected 0x0 default.
  function _addressNotNull(address _to) private pure returns (bool) {
    return _to != address(0);
  }

  /// For checking approval of transfer for address _to
  function _approved(address _to, uint256 _tokenId) private view returns (bool) {
    return gameItemIndexToApproved[_tokenId] == _to;
  }

  /// For creating gameItem
  function _createGameItem(string _name, address _owner, uint256 _price, uint _gameId) private {
    GameItem memory _gameItem = GameItem({
      name: _name,
      gameId: _gameId
    });
    uint256 newGameItemId = gameItems.push(_gameItem) - 1;

    // It's probably never going to happen, 4 billion tokens are A LOT, but
    // let's just be 100% sure we never let this happen.
    require(newGameItemId == uint256(uint32(newGameItemId)));

    Birth(newGameItemId, _name, _owner);

    gameItemIndexToPrice[newGameItemId] = _price;

    // This will assign ownership, and also emit the Transfer event as
    // per ERC721 draft
    _transfer(address(0), _owner, newGameItemId);
  }

  /// Check for token ownership
  function _owns(address claimant, uint256 _tokenId) private view returns (bool) {
    return claimant == gameItemIndexToOwner[_tokenId];
  }

  /// For paying out balance on contract
  function _payout(address _to) private {
    if (_to == address(0)) {
      ceoAddress.transfer(this.balance);
    } else {
      _to.transfer(this.balance);
    }
  }

  /*
  This function can be used by the owner of a GameItem item to modify the price of its GameItem item.
  */
  function modifyGameItemPrice(uint _gameItemId, uint256 _newPrice) public {
      require(_newPrice > 0);
      require(gameItemIndexToOwner[_gameItemId] == msg.sender);
      gameItemIndexToPrice[_gameItemId] = _newPrice;
  }

  /// @dev Assigns ownership of a specific gameItem to an address.
  function _transfer(address _from, address _to, uint256 _tokenId) private {
    // Since the number of gameItem is capped to 2^32 we can't overflow this
    ownershipTokenCount[_to]++;
    //transfer ownership
    gameItemIndexToOwner[_tokenId] = _to;

    // When creating new gameItems _from is 0x0, but we can't account that address.
    if (_from != address(0)) {
      ownershipTokenCount[_from]--;
      // clear any previously approved ownership exchange
      delete gameItemIndexToApproved[_tokenId];
    }

    // Emit the transfer event.
    Transfer(_from, _to, _tokenId);
  }

}
library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    if (a == 0) {
      return 0;
    }
    uint256 c = a * b;
    assert(c / a == b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  /**
  * @dev Substracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }

}


contract CryptoVideoGames {
    // This function will return only the owner address of a specific Video Game
    function getVideoGameOwner(uint _videoGameId) public view returns(address) {
    }
    
}